
10 Working With Mapping Services

106160 Tandem Computers Incorporated 10–1

This section discusses the service provided by NonStop NET/MASTER Management
Services (MS) called Mapping Services.  This service enables the manipulation of data
by NCL processes regardless of structure or format.  The section discusses the three
components in Mapping Services processing—maps, mapped data object (MDO)
variables, and NCL procedures.

This section does not discuss the structure of standard maps or how to create and use a
user-defined map.  These topics are discussed in Section 11, “Standard and
User-Defined Maps.”

Table 10-1 summarizes the verbs discussed in this section.  These are verbs that
support access to Mapping Services and data in MDO variables.

Table 10-1.  Verbs Used With Maps and MDO Variables

Verbs Description

ASSIGN verb Manipulates multiple variables in a single operation.

DROP verb Deletes (drops) multiple variables in a single operation.

EMS* verbs Controls the flow of Event Management Service (EMS) messages through the
system-level EMSPROC NCL procedure.

FILE verbs Opens and closes user databases (UDBs) and manipulates UDB records.

INT* verbs Controls the flow of messages sent from dependent processing environments.

LOG* verbs Controls the flow of messages through the system-level LOGPROC NCL
procedure.  These are messages destined for the NonStop NET/MASTER MS
activity log.

MSG* verbs Controls the flow of messages through the system-level MSGPROC NCL
procedure.  These are messages destined for the OCS window with which the
MSGPROC procedure is associated.

SECCALL verbs Manipulates user ID definition records in the User ID Management Services
(UMS) database.

WRITE verb Writes a message to a terminal, an NCL process, all monitor-class users, a
specified user, or the activity log.

INT*, EMS*, LOG*, and MSG* refer to all verbs beginning with the letters INT, EMS, LOG, and MSG,
respectively.

For the complete syntax of verbs, refer to the NonStop NET/MASTER NCL Reference
Manual.  (In this manual, Section 2, “NCL Summary,” discusses the meaning of the
asterisk (*) following the letters EMS, INT, LOG, and MSG.)

Components in
Mapping Services

Processing

There are three main components in Mapping Services processing.  The components
are briefly discussed here and discussed in detail throughout this section.

Maps
Mapped data object (MDO) variables
NCL procedures

Maps describe the structure of MDO variables.  MDO variables contain data.  The data
in an MDO variable is accessed from NCL procedures using certain verbs.

Mapping Services provides a set of facilities that mediates between these three
components.  This allows each component to exist as a separate and distinct entity in
NonStop NET/MASTER MS.  It allows Mapping Services to transparently manage
data manipulation as the components interact during the execution of an NCL process.

Maps Fundamental to Mapping Services is the concept of a map.  A map is a series of
definitions that, together, describe the entire structure of some item of data contained
in an MDO variable.  Each definition describes a single structure or substructure
within a data item.  Included in every definition is the following:

The unique symbolic name of the definition—this is part of the header of the
definition

The unique key value of the definition—this is part of the header of the definition

The unique symbolic name of every structure within the definition

The unique key of every structure within the definition

The unique symbolic name of every substructure within the definition

The unique key of every substructure within the definition

The position—or offset—at which every structure and substructure begins

The length—fixed or variable—of every structure and substructure

Mapping Services supports a map file that contains the definitions of all maps that
Mapping Services can use.  The map file and the maps $MSG, $NCL, $SEC, and $SPI
are discussed in detail in Section 11, “Standard and User-Defined Maps.”

Mapped Data Object (MDO)
Variables

An MDO variable contains data that is structured in a way that Mapping Services can
understand using a map.  All structures and substructures in an MDO variable have
names.  NCL can use the names to locate data at a certain position in an MDO variable.

The main data structure in an MDO variable is called an element.  Logical
substructures in an element are called groups and fields.  Understanding map
definitions allows you to understand and interpret the structure of data in an MDO
variable.  It allows you, during the execution of an NCL process, to locate and refer to
the data contained in elements, groups, and fields by name.  Map definitions provide
the names you use to refer to elements, groups, and fields from an NCL process.

The structures and substructures in an MDO variable are discussed in detail later in
this section.

Note An MDO variable can be either a stem or compound variable.  Refer to the discussion on variables in the
NonStop NET/MASTER NCL Reference Manual for the syntax of variables in general and stem and
compound variables in particular.

NCL Procedures You can write NCL procedures that access and manipulate the data contained in MDO
variables.  You access the data in MDO variables in the same way as you access the
data in other compound variables—by naming the segment in the variable that
contains the data you want to access.

Example of Mapping
Services Processing

The following two NCL procedures show how you can easily access the data in an
MDO variable.

The first NCL procedure is a MSGPROC procedure called ZEX1001N.  It is designed to
trap messages that begin with TEST and to analyze and display their contents—the
contents of messages are stored in MDO variables.

The second NCL procedure, called ZEX1002N, sends a message, which travels in an
MDO variable, to the OCS window from which ZEX1002N is executed.

To see how these NCL procedures work, follow these steps:

1. Type the first NCL procedure as shown, calling it ZEX1001N:

zex1001n: PROCEDURE
   /* MSGPROC procedure to analyze MDO variables */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         SAY "&$MSG.TEXT is "&$msg.text
         SAY "&$MSG.MSGATTR.DISPLAY.COLOR is "&$msg.msgattr.display.color
         SAY "&$MSG.MSGATTR.DISPLAY.HLITE is "&$msg.msgattr.display.hlite
         SAY "&$MSG.MSGATTR.DISPLAY.INTENS is "&$msg.msgattr.display.intens
         SAY "&$MSG.MSGATTR.ALARM is "&$msg.msgattr.alarm
         SAY "&$MSG.MSGATTR.FLAGS.NRD is "&$msg.msgattr.flags.nrd
         SAY "&$MSG.DOMID is "&$msg.domid
         SAY "&$MSG.SOURCE.TIME is "&$msg.source.time
         SAY "&$MSG.SOURCE.USER is "&$msg.source.user
         DO &i = 1 TO &SYS.VARCNT
            SAY "Word "&i" of text is "&word&i
            END
         END
      ELSE
         MSGCONT
   END
END zex1001n

2. Install ZEX1001N as your active MSGPROC procedure using the following
command from the OCS command input line:

PROFILE MSGPROC=ZEX1001N

3. Type the second NCL procedure as shown, calling it ZEX1002N:

zex1002n: PROCEDURE
   /* Sends a single message to an OCS window */
   WRITE COLOR=WHITE,
         MON=YES,
         NRD=OPER,
         DATA=TEST An OPER NRD message
END zex1002n

4. Execute ZEX1002N from the OCS command input line by using the following
command:

START ZEX1002N

5. Observe the results of installing ZEX1001N as your active MSGPROC procedure
and executing ZEX1002N, as shown in the following screen:

  (06:39) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1001N                                               <--  1
   NNM0393 MSGPROC PROCESSING ACTIVATED                                   <--  2
   NNM0357 PROFILE HAS CHANGED                                            <--  3
   START ZEX1002N                                                         <--  4
   TEST An OPER NRD message                                               <--  5
   &$MSG.TEXT is TEST An OPER NRD message                                 <--  6
   &$MSG.MSGATTR.DISPLAY.COLOR is WHITE                                   <--  7
   &$MSG.MSGATTR.DISPLAY.HLITE is NONE                                    <--  8
   &$MSG.MSGATTR.DISPLAY.INTENS is NORMAL                                 <--  9
   &$MSG.MSGATTR.ALARM is 0                                               <-- 10
   &$MSG.MSGATTR.FLAGS.NRD is 1                                           <-- 11
   &$MSG.DOMID is                                                         <-- 12
   &$MSG.SOURCE.TIME is 1992/06/15 06:39:09.388                           <-- 13
   &$MSG.SOURCE.USER is NMTJN                                             <-- 14
   Word 1 of text is TEST                                                 <-- 15
   Word 2 of text is An                                                   <-- 16
   Word 3 of text is OPER                                                 <-- 17
   Word 4 of text is NRD                                                  <-- 18
   Word 5 of text is message                                              <-- 19
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001005

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

In the following discussion, line numbers refer to the numbers at the right of the
preceding screen:

Line 1 shows the PROFILE MSGPROC command.

Lines 2 and 3 show the result of executing the PROFILE MSGPROC command.

Line 4 shows the START command.

The ZEX1001N NCL procedure uses the MSGREAD verb to read the message sent
by the ZEX1002N NCL procedure.  Line 5 shows the result of displaying the
message text on the OCS window.

When the MSGREAD verb reads a message, it automatically places the text and
the attributes of the text into an MDO variable beginning with the stem &$MSG..

This variable is called a message profile variable.  The names of all the structures
and substructures contained in the MDO variable begin with this stem.  Lines 6
through 14 show the result of using the SAY core statement to name and display
the contents of some of the substructures contained in the MDO variable.
Mapping Services uses the map $MSG to interpret the contents of the MDO
variable.

Note The value of &$MSG.SOURCE.TIME and &$MSG.SOURCE.USER may be different from the values
displayed here.

The names and contents that are displayed are summarized in the following table:

MDO Variable Name Contents

&$MSG.TEXT TEST An OPER NRD message

&$MSG.MSGATTR.DISPLAY.COLOR WHITE

&$MSG.MSGATTR.DISPLAY.HLITE NONE

&$MSG.MSGATTR.DISPLAY.INTENS NORMAL

&$MSG.MSGATTR.ALARM 0

&$MSG.MSGATTR.FLAGS.NRD 1

&$MSG.DOMID Null

&$MSG.SOURCE.TIME 1992/06/15 06:39:09.388

&$MSG.SOURCE.USER NMTJN

When the MSGREAD verb reads a message, it has the option of placing the text
(but not the attributes of the text) into ordinary variables using keywords such as
ARGS, VARS, RANGE, SEGMENT, and PARSE.  The VARS keyword places each
word of the text into a variable beginning with &WORD—&WORD1, &WORD2,
and so on.  Lines 15 through 19 show the result of using the SAY core statement to
display the contents of the variables that hold the text of the message.

Note Do not confuse the name of the MDO variable &$MSG. with the name of the map $MSG.  &$MSG. is a
variable created by the MSGREAD verb to contain the contents of a message.  The map $MSG is used to
interpret messages.  The map $MSG is also used to interpret the contents of MDO variables created by
the EMSREAD, INTREAD, and LOGREAD verbs—&$EMS., &$INT., and &$LOG., respectively.

The variables created by the EMSREAD, INTREAD, LOGREAD, and MSGREAD verbs are called
message profile variables.  Refer to the NonStop NET/MASTER NCL Reference Manual for the complete
set of values that are allowed in the fields of message profile variables.

The Structure of
Mapped Data Object

(MDO) Variables

The data item described by a map is stored in an MDO variable.  In general, an MDO
variable consists of a header and a single data structure, called an element, where an
element consists of a header followed by a data component.  The data component may
consist of other (enclosed) elements inside an outer (enclosing) element or other data
structures called groups and fields.  Some enclosed elements can be repeated and are
called repeated elements.  Data can be fixed or can vary in length.

This subsection discusses the structure of MDO variables and how to access the data
contained in MDO variables from an NCL procedure.  This subsection examines the
following topics:

The header of an MDO variable:

The length field
The key field
Referring to a whole MDO variable from NCL

Elements in an MDO variable

The header of an element:

The length field
The key field
Referring to an element from NCL

The data component of an element (and how to access data from NCL):

Enclosing and enclosed elements
Repeated elements
Fixed length data sections
Variable length data sections
Groups in elements
Fields in elements
Enumerated data types

The Header of an MDO
Variable

All MDO variables have a header.  An MDO variable header identifies the beginning
of an MDO variable.  A header consists of two parts:  a length field and a key field.
Figure 10-1 shows the structure of the header in an MDO variable mapped by the map
$MSG.

Figure 10-1.  Header in an MDO Variable

The Length Field of an MDO Variable Header

The length field of an MDO variable header contains the length of the whole MDO
variable.  Messages vary in length.  Accordingly, the value of this field can vary from
MDO variable to MDO variable.  This is reflected by the letters "len" in Figure 10-1.

The length of the length field itself is included in the length of an MDO variable.
Figure 10-1 shows this is fixed at 4 bytes in an MDO variable mapped by the map
$MSG.

When an MDO variable is created, Mapping Services automatically calculates the total
length of the variable and places it in the length field of the MDO variable header.

The length field of an MDO variable header contains internal Mapping Services data.
You cannot directly access the length field of an MDO variable header from an NCL
procedure.  However, the following NCL procedure shows how you can use the
LENGTH built-in function to indirectly access the length of an MDO variable:

zex1003n: PROCEDURE
   /* MSGPROC procedure to obtain MDO length */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
      /* Get length and add 8 bytes */
         &length = LENGTH(&$MSG.) + 8
         SAY "Length of &$MSG. is "&length" bytes "
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1003n

The NCL procedure adds 8 bytes to the value obtained by the LENGTH built-in
function to include the length of the MDO variable header as part of the total length of
the MDO variable.  Using this technique requires an understanding of the structure of
the MDO variable.

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (06:49) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1003N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Length of &$MSG. is 187 bytes
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001009
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The Key Field of an MDO Variable Header

The key field of an MDO variable header uniquely identifies the map used to access
the MDO variable.  The value contained in the key field is the name of the map that
Mapping Services uses to interpret the MDO variable.  The name of the map in
Figure 10-1 is $MSG.  This is called a symbolic name.

The symbolic name of a map is represented internally in an MDO variable.  The
internal representation of a symbolic name is called a key value.  Figure 10-1 shows
the key value for $MSG is (the decimal number) 2.  In an MDO variable, when
Mapping Services encounters the key value of 2 as part of the MDO variable header,
Mapping Services interprets it to mean:  "Use the map $MSG when analyzing this
MDO variable."  Every MDO variable described by the map $MSG has the key value of
2 as part of its header.

The length of the key field is included in the length of an MDO variable.  Figure 10-1
shows this is fixed at 4 bytes in an MDO variable.

When an MDO variable is created, Mapping Services automatically encodes the name
of the map used to interpret the MDO variable in the key field of the MDO variable
header.  When you refer to an MDO variable from an NCL procedure, Mapping
Services uses the map name to interpret the structures within the variable.

The key field of an MDO variable header contains internal Mapping Services data.
You cannot directly access the key field of an MDO variable header from an NCL
procedure.  You can use the SHOW MAPS command to display the symbolic names
and corresponding key values of map definitions, as the following screen shows:

  (07:34) --------------------- OPERATOR CONTROL SERVICES ----------------------
   SHOW MAPS
   NNM0256 MAP-NAME     MAP-KEY   USE-CNT  STATUS
   NNM0257 $MAP         00000030        0  NAMEONLY
   NNM0257 $MSG         00000002        0  LOADED
   NNM0257 $NCL         00000001        0  LOADED
   NNM0257 $SEC         00000010        1  LOADED
   NNM0257 $SPI         00000077        1  LOADED
   NNM0999 *END* 5 MAPS DISPLAYED
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Referring to a Whole MDO Variable From NCL

A stem variable is a simple or complex variable that ends with a period (.).  A stem is
used as a prefix for compound variables.  Syntactically, MDO variables are compound
variables.  You use the stem of an MDO variable from an NCL procedure to refer to
the whole MDO variable.  Referring to the stem refers to all compound variables under
the stem.

The following example refers to the whole MDO variable that is created by the
MSGREAD verb when it reads a message—notice the trailing period (.):

&$MSG.

Note You must distinguish between the following two variable names.  The first variable does not have a trailing
period and is a simple variable, not a stem variable.  The second variable has a trailing period and is a
stem variable;  it may be the stem of an MDO variable.

&$MSG  /* No trailing period—a simple variable */
&$MSG. /* Has a trailing period—a stem variable */

The following MSGPROC NCL procedure refers to the stem of an MDO variable:

zex1004n: PROCEDURE
   /* MSGPROC procedure to analyze MDO stem */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         SAY "&$MSG is "&$msg
         SAY "&$MSG. is "&$msg.
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1004n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (06:52) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1004N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   &$MSG is
   &$MSG. is  . .TEST An OPER NRD
   message 9 ....   ..   .    .. ".        .           ..           ^ ...1992/06
   /15 06:52:01.090....SYS1..SYS1    ....SYS1..SYS1    ..NMTJN
   ..#9446690..  ....
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001012
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The screen shows that the simple variable, &MSG, has a null value and that the stem
of the MDO variable, &$MSG., refers to the data in the whole MDO variable.

This example also shows that it is generally not useful to display the contents of a
whole MDO variable, such as &$MSG..  As a general rule, it is only useful to refer to a
whole MDO variable to copy its contents to another variable.  Referring to a whole
MDO variable allows you to copy large blocks of data from one variable to another.
The syntax of the ASSIGN verb that allows you to copy MDO variables is discussed
later in this section.

Elements in an MDO
Variable

An MDO variable is composed of one or more data structures called elements.  An
element is the major logical data structure that Mapping Services understands.  An
element consists of a header followed by a data component.  The data component may
consist of other (enclosed) elements inside an outer (enclosing) element or other data
structures called groups and fields.  Some enclosed elements can be repeated and are
called repeated elements.

The Header of an Element The header of an element identifies the beginning of an element in an MDO variable.
An element header consists of two parts:  a length field and a key field.  Figure 10-2
shows the structure of the header of an element called MSGATTR inside an MDO
variable mapped by $MSG.

Figure 10-2.  Header of an Element in an MDO Variable

The Length Field of an Element Header

The length field of an element in an MDO variable contains the length of the whole
element.  The length may be fixed, in which case the length is always known, or it may
be variable.  The length is fixed if there are no variable length structures in the
element.  The length is variable if the element contains one or more variable length
structures.  The letters "len" in Figure 10-2 indicate that the length of the MSGATTR
element in an MDO variable mapped by the map $MSG is variable.

The length of the length field itself is included in the length of an element.  Figure 10-2
shows this is fixed at 2 bytes for the MSGATTR element in an MDO variable mapped
by the map $MSG.

When an MDO variable is created, Mapping Services automatically calculates the total
length of each element and places it in the length field of the element header.



The Structure of Mapped Data Object (MDO) Variables

Working With Mapping Services

106160 Tandem Computers Incorporated 10–13

The length field of an element header contains internal Mapping Services data.  You
cannot directly access the length field of an element header from an NCL procedure.
However, the following NCL procedure shows how you can use the LENGTH built-in
function to indirectly access the length of an element:

zex1005n: PROCEDURE
   /* MSGPROC procedure to obtain element length */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
      /* Get length and add 4 bytes */
         &length = LENGTH( &$MSG.TEXT ) + 4
         SAY "Length of &$MSG.TEXT is "&length" bytes "
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1005n

The NCL procedure adds 4 bytes to the value obtained by the LENGTH built-in
function to include the length of the element header as part of the total length of the
element.  Using this technique requires an understanding of the data structure of an
element.

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (06:54) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1005N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Length of &$MSG.TEXT is 28 bytes
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001015
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The Key Field of an Element Header

The key field of an element in an MDO variable uniquely identifies the element.  The
value contained in the key field represents the encoded name of the element definition
that Mapping Services uses to interpret the element.  The name of the definition in
Figure 10-2 is MSGATTR.  This is called a symbolic name.

The symbolic name of an element definition is represented internally in an MDO
variable.  The internal representation of a symbolic name is called a key value.  Figure
10-2 shows the key value for MSGATTR is (the decimal number) 2.  In the element of
an MDO variable mapped by the map $MSG, when Mapping Services encounters the
key value of 2, Mapping Services interprets it to mean:  "Use the definition for the
element MSGATTR when analyzing this element."

The length of the key field is included in the length of an element.  Figure 10-2 shows
this is fixed at 2 bytes in MSGATTR.

When an MDO variable is created, Mapping Services automatically inserts the name of
the definition used to interpret an element in the key field of the element header.
When you refer to an element from an NCL procedure, Mapping Services uses the
name to interpret the structures within the element.

The key field of an element header contains internal Mapping Services data.  You
cannot directly access the key field of an element header from an NCL procedure.
However, the following NCL procedure shows how you can use the C2D (returns the
decimal value of a string) and SUBSTR (returns a substring) built-in functions to
indirectly access the key value of an element:

zex1006n: PROCEDURE
   /* MSGPROC procedure to obtain element key value */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         &key_value = C2D( SUBSTR( &$MSG., 3, 2 ) )
         SAY "Key value of &$MSG.TEXT is "&key_value
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1006n

Using this technique requires an understanding of the structure of the element.



The Structure of Mapped Data Object (MDO) Variables

Working With Mapping Services

106160 Tandem Computers Incorporated 10–15

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (06:56) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1006N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Key value of &$MSG.TEXT is 1
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001018
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Referring to an Element From NCL

You can refer to the symbolic name of an element in an MDO variable from an NCL
procedure.  Doing so, uniquely identifies and refers to the value of the element.

The following example refers to the MSGATTR element in an MDO variable that is
created by the MSGREAD verb when it reads a message:

&$MSG.MSGATTR

Note The trailing period after an element name in an MDO variable is effectively ignored by Mapping Services.
Mapping Services regards the following two variable names as the same:

&$MSG.MSGATTR  /* No trailing period */
&$MSG.MSGATTR. /* Has a trailing period */

The following MSGPROC NCL procedure refers to an element in an MDO variable:

zex1007n: PROCEDURE
   /* MSGPROC procedure to analyze elements */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         SAY "&$MSG.MSGATTR is "&$msg.msgattr
         SAY "&$MSG.MSGATTR. is "&$msg.msgattr.
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1007n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (06:58) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1007N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   &$MSG.MSGATTR is ...   ..   .    .. ".        .           ..
   &$MSG.MSGATTR. is ...   ..   .    .. ".        .           ..
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001021
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The screen shows that the two MDO variables that refer to the value of the MSGATTR
element produce the same result.

This example also shows, as with whole MDO variables, that it is generally not useful
to display the contents of an enclosed element such as MSGATTR.  As a general rule, it
is only useful to refer to an enclosing element to copy its contents to another variable.
Referring to an enclosing element allows you to copy large blocks of data from one
variable to another.  The syntax of the ASSIGN verb that allows you to do this is
discussed later in this section.  Enclosed and enclosing elements are discussed later in
this section.

The Data Component of an
Element

The data component of an element follows the header of an element.  The component
is conceptually divided into two sections:  a fixed length data section and a variable
length data section.  Either section may contain a null value.  The data component may
consist of other (enclosed) elements inside an outer (enclosing) element or other data
structures called groups and fields.  Some enclosed elements can be repeated and are
called repeated elements.

The data component contains data that you can access from an NCL procedure.  The
data component which you generally refer to from an NCL procedure is a field.

Figure 10-3 shows part of the structure of an element called MSGATTR in an MDO
variable mapped by the map $MSG.

Enclosing and Enclosed Elements

Some elements—called enclosing elements—may contain other elements—called
enclosed elements.  Figure 10-3 shows some of the elements in the MSGATTR element
in an MDO variable mapped by the map $MSG:  DISPLAY, FLD and MSGCODE.
MSGATTR is an enclosing element while DISPLAY, FLD, and MSGCODE are enclosed
elements.

An enclosed element is contained entirely within the data component of the enclosing
element.  Its length is included as part of the value of the length field of the enclosing
element.  Figure 10-3 shows that the elements DISPLAY, FLD and MSGCODE are part
of the data component of MSGATTR.  Figure 10-3 also shows that the elements
DISPLAY, FLD and MSGCODE have their own data components.

Since elements can contain further elements, it follows that a hierarchy of elements can
exist.  An enclosed element is subordinate in the hierarchy to the element that encloses
it.  For example, DISPLAY is subordinate to MSGATTR, FLD is subordinate to
MSGATTR, and MSGCODE is subordinate to MSGATTR.

Figure 10-3.  Partial Structure of an Element in an MDO Variable

MSGATTR
(02)

2 byte

len

2 byte

Length
Field

Key
Field

Header Data Component

Element MSGATTR

MSGCODE
 (24)

1 byte

msgcode
 

1 byte

3
 

1 byte

Length
Field

Key
Field

Header Data
Field

Element MSGCODE

DISPLAY
 (03)

1 byte

COLOR
 COLOR#

1 byte

HLITE
 HLITE#
1 byte

INTENS
 INTENS#

1 byte

BKGCOLOR
 BKGCOLOR#

1 byte

6
 

1 byte

Length
Field

Key
Field

Header Data Component

Four Fields

Element DISPLAY

FLD
 (04)

1 byte

OFFSET
 

2 byte

LENGTH
 

2 byte

10
 

1 byte

COLOR
 COLOR#

1 byte

HLITE
 HLITE#
1 byte

INTENS
 INTENS#

1 byte

BKGCOLOR
 BKGCOLOR#

1 byte

Length
Field

Key
Field

Header Data Component

Element FLD

Six Fields

Element
can be

repeated

006



The Structure of Mapped Data Object (MDO) Variables

Working With Mapping Services

106160 Tandem Computers Incorporated 10–19

Each element inside an enclosing element is interpreted as another element.  It has a
symbolic name provided by the map that is used to interpret it.  Symbolic names
among elements at the same level must be unique.

Since elements are hierarchical and symbolic names are unique at the same level, you
can uniquely identify and refer to an enclosed element from an NCL procedure.  To do
so, you must name all intermediate elements in the hierarchy, as the following
examples show:

&$EMS.MSGATTR                 /* Refers to MSGATTR element */
&$INT.MSGATTR.DISPLAY         /* Refers to DISPLAY element */
&$LOG.MSGATTR.FLD             /* Refers to FLD element     */
&$MSG.MSGATTR.MSGCODE         /* Refers to MSGCODE element */

Repeated Elements

Some elements can be repeated.  An element that can be repeated is called a repeated
element.  Figure 10-3 shows that the FLD element in an MDO variable mapped by the
map $MSG is a repeated element.

Figure 10-4 shows part of the hierarchy of the map $MSG.

Figure 10-4.  Part of the Hierarchy of $MSG

Note The main display attributes of the text of a message are stored in the DISPLAY element of an MDO
variable.  The FLD element is used to assign different attributes to parts of the text.  If the entire text of a
message has the same display attributes, the FLD element is not present in the MDO variable.

When you are referring to a repeated element from NCL, you must use an integer
index to uniquely identify the specific instance of the repeated element.  An integer
index is a number, or an expression that resolves to a number, enclosed in braces and
placed after the symbolic name of an element.  The value of the integer index indicates
the occurrence of the element to which you want to refer.

The following examples refer to specific repeated elements:

&$EMS.MSGATTR.FLD{1}          /* Refers to 1st FLD element */
&$INT.MSGATTR.FLD{2}          /* Refers to 2nd FLD element */
&$LOG.MSGATTR.FLD{3}          /* Refers to 3rd FLD element */
&$MSG.MSGATTR.FLD{4}          /* Refers to 4th FLD element */

Note The {1} is not required for the first occurrence of a repeated element;  this is the default.  The following
two variable names both refer to the first repeated element:

&$MSG.MSGATTR.FLD
&$MSG.MSGATTR.FLD{1}

The following example refers to the repeated element whose value is determined by
the value of the simple variable, &INDEX, within the braces:

&$EMS.MSGATTR.FLD{&index}

If the value of &INDEX is 12, for example, then the preceding example refers to the
following repeated element:

&$EMS.MSGATTR.FLD{12}

You can use an assignment statement to create and assign values to fields in repeated
elements as the following examples show:

&$MSG.MSGATTR.FLD{1}.OFFSET = 8
&$MSG.MSGATTR.FLD{1}.LENGTH = 8
&$MSG.MSGATTR.FLD{1}.COLOR = "TURQUOISE"
&$MSG.MSGATTR.FLD{1}.HLITE = "BLINK"
&$MSG.MSGATTR.FLD{1}.INTENS = "HIGH"

When creating and assigning values to fields in repeated elements, you do not have to
use consecutive integer indexes.  For example, you can assign values to fields in
&$MSG.MSGATTR.FLD{2} without first assigning values to fields in
&$MSG.MSGATTR.FLD{1}.

The following MSGPROC NCL procedure reads a message, assigns values to fields in
an element that can be repeated, and displays the modified message on an OCS
window:

zex1008n: PROCEDURE
   /* MSGPROC procedure to create a repeated element */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         &$msg.msgattr.fld{1}.offset = 8
         &$msg.msgattr.fld{1}.length = 8
         &$msg.msgattr.fld{1}.color = "TURQUOISE"
         &$msg.msgattr.fld{1}.hlite = "BLINK"
         &$msg.msgattr.fld{1}.intens = "HIGH"
         SAY "&$MSG.TEXT is "&$msg.text
         SAY "&$MSG.MSGATTR.DISPLAY.COLOR is "&$msg.msgattr.display.color
         SAY "&$MSG.MSGATTR.DISPLAY.HLITE is "&$msg.msgattr.display.hlite
         SAY "&$MSG.MSGATTR.DISPLAY.INTENS is "&$msg.msgattr.display.intens
         SAY "&$MSG.MSGATTR.FLD{1}.OFFSET is "&$msg.msgattr.fld{1}.offset
         SAY "&$MSG.MSGATTR.FLD{1}.LENGTH is "&$msg.msgattr.fld{1}.length
         SAY "&$MSG.MSGATTR.FLD{1}.COLOR is "&$msg.msgattr.fld{1}.color
         SAY "&$MSG.MSGATTR.FLD{1}.HLITE is "&$msg.msgattr.fld{1}.hlite
         SAY "&$MSG.MSGATTR.FLD{1}.INTENS is "&$msg.msgattr.fld{1}.intens
         SAY "&$MSG.MSGATTR.ALARM is "&$msg.msgattr.alarm
         SAY "&$MSG.SOURCE.TIME is "&$msg.source.time
         SAY "&$MSG.SOURCE.USER is "&$msg.source.user
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1008n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:01) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1008N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   &$MSG.TEXT is TEST An OPER NRD message
   &$MSG.MSGATTR.DISPLAY.COLOR is WHITE
   &$MSG.MSGATTR.DISPLAY.HLITE is NONE
   &$MSG.MSGATTR.DISPLAY.INTENS is NORMAL
   &$MSG.MSGATTR.FLD{1}.OFFSET is 8
   &$MSG.MSGATTR.FLD{1}.LENGTH is 8
   &$MSG.MSGATTR.FLD{1}.COLOR is TURQUOISE
   &$MSG.MSGATTR.FLD{1}.HLITE is BLINK
   &$MSG.MSGATTR.FLD{1}.INTENS is HIGH
   &$MSG.MSGATTR.ALARM is 0
   &$MSG.SOURCE.TIME is 1992/06/15 07:01:56.548
   &$MSG.SOURCE.USER is NMTJN
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001024
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The screen shows that the OPER NRD part of the text is displayed with attributes that
are different from the base part of the text.

Fixed Length Data Sections

A fixed length data section is a section of data where the length of the data is known.
Fixed length data sections of an element can consist of only groups and fields.
(Groups and fields are described in more detail later in this section.)  Figure 10-3
shows that the data component of the DISPLAY, FLD, and MSGCODE elements are of
fixed length.

Variable Length Data Sections

A variable length data section is a section of data where the length of the data is not
known before the data exists.  The length is variable if the element contains one or
more variable length structures.  Figure 10-3 shows that the data component of the
MSGATTR element is of variable length.

Groups in Elements

A group is any contiguous section of data within the fixed length data section of an
element.  The contents of the element can consist of one or more groups.  Each group
can be further decomposed into other groups or fields.

A group may have a fixed length or a variable length.  A variable length group extends
to the end of the group in which it is enclosed, or, if it is an outermost group, to the
end of the element in which it is defined.

Fields in Elements

A field is the most fundamental unit of an MDO variable.  Fields contain the basic
units of meaningful data in an MDO variable.

The start of a field has a fixed offset in a group and consists of a sequence of
contiguous bytes within that group.  A field can be fixed or variable in length.  It
extends to the end of the current group, as described previously for groups.

Note Data assigned to a fixed length field is truncated if it exceeds the length of the field.

Offsets begin from 0 (zero).  Figure 10-3 shows that value of COLOR in the element
DISPLAY begins at offset 0.  However, the value of COLOR in the element FLD begins
at offset 4 because the value of the fields OFFSET and LENGTH are 2 bytes each.

Note the difference between fields and groups.  A group is a structure that can contain
other groups or fields.  A field cannot contain other fields (or groups).  Other than this
distinction, groups and fields are processed in exactly the same way by Mapping
Services.

Fields have symbolic names, which must be unique among other field names at the
same hierarchical level.  You use these names to refer to the data defined within the
field.  The following examples refer to fields in the element MSGATTR:

&$EMS.MSGATTR.DISPLAY.COLOR           /* Base COLOR field  */
&$EMS.MSGATTR.DISPLAY.HLITE           /* Base HLITE field  */
&$EMS.MSGATTR.DISPLAY.INTENS          /* Base INTENS field */
&$INT.MSGATTR.FLD{1}.OFFSET           /* 1st OFFSET field  */
&$INT.MSGATTR.FLD{1}.LENGTH           /* 1st LENGTH field  */
&$INT.MSGATTR.FLD{1}.COLOR            /* 1st COLOR field   */
&$INT.MSGATTR.FLD{1}.HLITE            /* 1st HLITE field   */
&$INT.MSGATTR.FLD{1}.INTENS           /* 1st INTENS field  */
&$MSG.MSGATTR.FLD{2}.OFFSET           /* 2nd OFFSET field  */
&$MSG.MSGATTR.FLD{2}.LENGTH           /* 2nd LENGTH field  */
&$MSG.MSGATTR.FLD{2}.COLOR            /* 2nd COLOR field   */
&$MSG.MSGATTR.FLD{2}.HLITE            /* 2nd HLITE field   */
&$MSG.MSGATTR.FLD{2}.INTENS           /* 2nd INTENS field  */

The following MSGPROC NCL procedure reads a message, assigns values to fields in
an element that can be repeated, and displays the modified message on an OCS
window:

zex1009n: PROCEDURE
   /* MSGPROC procedure to display fields */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         &$msg.msgattr.fld{1}.offset = 8
         &$msg.msgattr.fld{1}.length = 8
         &$msg.msgattr.fld{1}.color = "TURQUOISE"
         &$msg.msgattr.fld{1}.hlite = "BLINK"
         &$msg.msgattr.fld{1}.intens = "HIGH"
         &$msg.msgattr.fld{2}.offset = 17
         &$msg.msgattr.fld{2}.length = 7
         &$msg.msgattr.fld{2}.color = "BLUE"
         &$msg.msgattr.fld{2}.hlite = "NONE"
         &$msg.msgattr.fld{2}.intens = "NORMAL"
         SAY "&$MSG.TEXT is "&$msg.text
         SAY "&$MSG.MSGATTR.DISPLAY.COLOR is "&$msg.msgattr.display.color
         SAY "&$MSG.MSGATTR.DISPLAY.HLITE is "&$msg.msgattr.display.hlite
         SAY "&$MSG.MSGATTR.DISPLAY.INTENS is "&$msg.msgattr.display.intens
         DO &i = 1 TO 2
           SAY "&$MSG.MSGATTR.FLD{"&i"}.OFFSET is "&$msg.msgattr.fld{&i}.offset
           SAY "&$MSG.MSGATTR.FLD{"&i"}.LENGTH is "&$msg.msgattr.fld{&i}.length
           SAY "&$MSG.MSGATTR.FLD{"&i"}.COLOR is "&$msg.msgattr.fld{&i}.color
           SAY "&$MSG.MSGATTR.FLD{"&i"}.HLITE is "&$msg.msgattr.fld{&i}.hlite
           SAY "&$MSG.MSGATTR.FLD{"&i"}.INTENS is "&$msg.msgattr.fld{&i}.intens
           END
         SAY "&$MSG.MSGATTR.ALARM is "&$msg.msgattr.alarm
         SAY "&$MSG.SOURCE.TIME is "&$msg.source.time
         SAY "&$MSG.SOURCE.USER is "&$msg.source.user
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1009n

The NCL procedure uses a DO loop to display the elements that are repeated—two in
this example—using a variable as the integer index.

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:06) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1009N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   &$MSG.TEXT is TEST An OPER NRD message
   &$MSG.MSGATTR.DISPLAY.COLOR is WHITE
   &$MSG.MSGATTR.DISPLAY.HLITE is NONE
   &$MSG.MSGATTR.DISPLAY.INTENS is NORMAL
   &$MSG.MSGATTR.FLD{1}.OFFSET is 8
   &$MSG.MSGATTR.FLD{1}.LENGTH is 8
   &$MSG.MSGATTR.FLD{1}.COLOR is TURQUOISE
   &$MSG.MSGATTR.FLD{1}.HLITE is BLINK
   &$MSG.MSGATTR.FLD{1}.INTENS is HIGH
   &$MSG.MSGATTR.FLD{2}.OFFSET is 17
   &$MSG.MSGATTR.FLD{2}.LENGTH is 7
   &$MSG.MSGATTR.FLD{2}.COLOR is BLUE
   &$MSG.MSGATTR.FLD{2}.HLITE is NONE
   &$MSG.MSGATTR.FLD{2}.INTENS is NORMAL
   &$MSG.MSGATTR.ALARM is 0
   &$MSG.SOURCE.TIME is 1992/06/15 07:06:06.450
 MSG QUEUED ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The screen shows that the “OPER NRD” part of the text and the “message” part of the
text are displayed with attributes that are different from the base part of the text.

Enumerated Data Types It is useful to divide MDO variables into two types—those that store raw data only
and those that store raw data that represents a symbolic name.  For convenience, the
former are called unenumerated MDO variables and the latter are called enumerated
MDO variables.

The ALARM field in a message profile variable stores raw data only—it is an
unenumerated MDO variable.  To access the data in the ALARM field, use the
following MDO variable name:

&$MSG.MSGATTR.ALARM

The COLOR field in a message profile variable stores raw data that represents a
symbolic name—it is an enumerated MDO variable.  In these variables, you can access
both the raw data that represents a symbolic name and the symbolic name itself in the
following way:

To obtain raw data, specify an MDO variable name with a number sign (#), for
example:

&$MSG.MSGATTR.DISPLAY.COLOR#

To obtain a symbolic name, specify an MDO variable name without a number
sign, for example:

&$MSG.MSGATTR.DISPLAY.COLOR

The following MSGPROC NCL procedure reads a message, assigns different colors to
the &$MSG.MSGATTR.DISPLAY.COLOR MDO variable, and displays both
unenumerated and enumerated MDO variables:

zex1010n: PROCEDURE
   /* MSGPROC procedure to display external and internal values */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         SAY "&$MSG.TEXT is "&$msg.text
         SAY "&$MSG.MSGATTR.DISPLAY.COLOR is "&$msg.msgattr.display.color
         SAY "&$MSG.MSGATTR.DISPLAY.COLOR# is "&$msg.msgattr.display.color#
         SAY "Changing colors"
         &$msg.msgattr.display.color = "BLUE"
         SAY &$MSG.MSGATTR.DISPLAY.COLOR" = "&$msg.msgattr.display.color#
         &$msg.msgattr.display.color = "RED"
         SAY &$MSG.MSGATTR.DISPLAY.COLOR" = "&$msg.msgattr.display.color#
         &$msg.msgattr.display.color = "GREEN"
         SAY &$MSG.MSGATTR.DISPLAY.COLOR" = "&$msg.msgattr.display.color#
         &$msg.msgattr.display.color = "TURQUOISE"
         SAY &$MSG.MSGATTR.DISPLAY.COLOR" = "&$msg.msgattr.display.color#
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1010n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:10) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1010N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   &$MSG.TEXT is TEST An OPER NRD message
   &$MSG.MSGATTR.DISPLAY.COLOR is WHITE
   &$MSG.MSGATTR.DISPLAY.COLOR# is -9
   Changing colors
   BLUE = -15
   RED = -14
   GREEN = -12
   TURQUOISE = -11
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001031
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Refer to the NonStop NET/MASTER NCL Reference Manual for the complete set of
values that are allowed in the fields of message profile variables.



Verbs and MDO Variable Access

Working With Mapping Services

10–28 106160 Tandem Computers Incorporated

Verbs and MDO
Variable Access

Verbs manipulate data using variables.  When a verb accesses data in a variable, it uses
keywords that support variable access—the ARGS, VARS, RANGE, GENERIC,
BYNAME, MERGE, OVERLAY, DATA, MDO, and MAP keywords.  MDO data is
stored in MDO variables.  You use the MDO and MAP keywords in verbs to access
MDO variables and maps.

Note Refer to the discussion on verb syntax and variable access methods in the NonStop NET/MASTER NCL
Reference Manual for information on verb keywords that are used to access variables other than MDO
variables—the ARGS, VARS, RANGE, GENERIC, BYNAME, MERGE, OVERLAY, and DATA keywords.

The MDO Operand

The MDO operand is used by verbs to explicitly name the stem of an MDO variable.
The syntax of the MDO operand is:

MDO=stem-name.

stem-name.

specifies the name of any user-defined stem variable that you want to contain
MDO data.

When you name the stem variable, you can give it any valid stem variable name on the
first occasion you use it (normally in the ASSIGN verb when you create the MDO
variable).  When you refer to the stem variable on subsequent occasions, NCL
understands that it is an MDO variable and that it must use Mapping Services to
interpret the data.  Referring only to the stem of an MDO variable refers to the data
contained in all elements that begin with the stem.

The syntax description shows that MDO variables look like stem variables.  You can
think of MDO variables as stem variables from the point of view of variable
substitution.  However, the structure of the data contained in an MDO variable means
that the data must be interpreted by Mapping Services.

With one exception, you must always follow the MDO operand with a stem variable
specification.  The exception is after the MDO operand on the ASSIGN verb where, in
one syntax format, you can follow the MDO operand with a compound variable
specification.  The ASSIGN verb is discussed later in this section.

MDO variables have certain restrictions.  An MDO variable cannot be a global
variable.  Its scope is restricted to the NCL process that creates it.  However, it can be
shared to called procedures or functions in a share list.

The following examples use the MDO operand to specify MDO variables:

ASSIGN MDO=&abcdef. MAP=$SEC
ASSIGN MDO=&XYZ_1.
ASSIGN MDO=&MY_VAR. FROM MDO=&$INT.
EMSALERT MDO=&my_ems.
FILE ADD ID=INPUT_1 MDO=&record.
FILE GET ID=FILE1 MDO=&file_rec.
SECCALL UPDATE USERID=NNMABC MDO=&ums_record.

The MAP Operand

The MAP operand is used by some verbs to specify the name of the map used to
analyze data in an MDO variable.  The syntax of the MAP operand is:

MAP=map-name

map-name

specifies the name of the map you want Mapping Services to use to interpret the
data in an MDO variable.  The map name is usually one of $MSG, $NCL, $SEC, or
$SPI.

The following examples use the MAP operand to specify the name of a map:

ASSIGN MDO=&abcdef. MAP=$SEC
ASSIGN MDO=&message. MAP=$MSG

The ASSIGN and
DROP Verbs

You can use the ASSIGN and DROP verbs to work with MDO variables.  The ASSIGN
verb has four syntax formats.  You can use two of these syntax formats to work with
MDO variables:

Format 1 of the ASSIGN verb allows you to manipulate multiple variables,
including MDO variables, in a single operation.

Note This section discusses format 1 of the ASSIGN verb in relation to MDO variables only.  Refer to the
discussion on verb syntax and variable access methods in the NonStop NET/MASTER NCL Reference
Manual for a discussion of format 1 of the ASSIGN verb in relation to ordinary variables.

Format 2 of the ASSIGN verb allows you to obtain information about MDO
variables and maps.

Note Format 2 of the ASSIGN verb is used to obtain information about variables, not to assign values to
variables.

The DROP verb has one syntax format and is used to delete (drop) MDO variables.

Format 1 of the ASSIGN
Verb

Format 1 of the ASSIGN verb is the default form of the verb (where OPT=VALUE
follows the ASSIGN keyword).

Creating an MDO Variable

MDO variables can be created in two ways—automatically by Mapping Services as a
result of certain verbs or explicitly by using the ASSIGN verb.  Table 10-2 lists the
verbs that automatically create MDO variables.

Table 10-2.  Verbs That Automatically Create an MDO Variable

Verb Comment

EMSREAD Automatically creates an MDO variable to hold an Event Management
Service (EMS) message.  The stem is always &$EMS..

INTREAD Automatically creates an MDO variable to hold a message from a dependent
processing environment.  The stem is always &$INT..

FILE GET Automatically creates an MDO variable to hold a record read from a mapped
user database (UDB).  Naming a stem after the MDO operand on the FILE
GET verb creates an MDO variable with that stem name.

FILE PUTGET Automatically creates an MDO variable when it gets a record from a
Guardian process.  Naming a stem after the MDO operand in the FILE
PUTGET verb creates an MDO variable with that stem name.

LOGREAD Automatically creates an MDO variable to hold a message destined for the
activity log.  The stem is always &$LOG..

MSGREAD Automatically creates an MDO variable to hold a message destined for an
OCS window.  The stem is always &$MSG..

PSEND SEND Automatically creates an MDO variable to hold the reply message from a
Pathway server class.  Naming a stem after the MDO operand in the PSEND
SEND verb creates an MDO variable with that stem name.

SECCALL GET Automatically creates an MDO variable to hold a record read from the User
ID Management Services (UMS) database.  Naming a stem after the MDO
operand in the SECCALL GET verb creates an MDO variable with that stem
name.

SECCALL QUERY Automatically creates an MDO variable to hold user ID information read from
memory.  Naming a stem after the MDO operand in the SECCALL QUERY
verb creates an MDO variable with that stem name.

Table 10-3 lists the verbs that can use an MDO variable after you explicitly create the
MDO variable using the ASSIGN verb.  You must use the ASSIGN verb to create MDO
variables before use by the verbs listed in this table.

Table 10-3.  Verbs That Can Use an MDO Variable After the Variable Is Created

Verb Comment

EMSSEND Sends a message from NCL to the EMS collector using the stem name you
specify.

FILE ADD Adds a record to a mapped user database (UDB) using the stem name you
specify.

FILE PUT Adds a record to or replaces a record in a mapped user database (UDB)
using the stem name you specify.

FILE PUTGET Puts a record to an operating system process using the stem name you
specify.  (The PUT does not automatically create an MDO variable.)

PSEND SEND Sends a request message to a Pathway server class using the stem name
you specify.

SECCALL ADD Adds a record to the User ID Management Services (UMS) database using
the stem name you specify.

SECCALL CHANGE Changes a record in the User ID Management Services (UMS) database
using the stem name you specify.

SECCALL UPDATE Updates a record in the User ID Management Services (UMS) database
using the stem name you specify.

START Sends a message to the started NCL process using the stem name you
specify.

WRITE Can send an MDO variable to another NCL process using the stem name
you specify.

If you want to construct an MDO variable, use the ASSIGN verb with the following
syntax:

ASSIGN [ OPT=VALUE ] MDO=variable-name MAP=map-name

MDO=variable-name

specifies the stem name of an MDO variable.

MAP=map-name

specifies the name of the map to be used by Mapping Services to interpret the data
in the MDO variable when it is accessed by an NCL process.

The following NCL procedure uses the ASSIGN verb to create an MDO variable,
&ABC., which is mapped by the map $MSG:

zex1011n: PROCEDURE
   /* Creates an MDO variable using ASSIGN */
   SAY Creating an MDO variable with map $MSG
   ASSIGN MDO=&abc. OPT=VALUE MAP=$MSG
   SAY Creating fields in the MDO variable
   &abc.text                   = "A test message"
   &abc.msgattr.display.color  = RED
   &abc.msgattr.display.hlite  = NONE
   &abc.msgattr.display.intens = NORMAL
   SAY Displaying fields in the MDO variable
   SAY "Text is "&abc.text
   SAY "Color is "&abc.msgattr.display.color
   SAY "Hlite is "&abc.msgattr.display.hlite
   SAY "Intens is "&abc.msgattr.display.intens
END zex1011n

The NCL procedure creates a series of compound variables under the stem &ABC.
using the assignment statement to assign values to each.

The following screen shows the results of executing the procedure:

  (14:50) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1011N
   Creating an MDO variable with map $MSG
   Creating fields in the MDO variable
   Displaying fields in the MDO variable
   Text is A test message
   Color is RED
   Hlite is NONE
   Intens is NORMAL
   NNM1005 START ZEX1011N PROCESSING COMPLETE. NCLID 001548
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Deleting an MDO Variable

You can delete an MDO variable in two ways:  using the DROP verb or using the
ASSIGN verb.  The following example of the DROP verb deletes an MDO variable:

DROP VARS=&abc.

The following syntax of the ASSIGN verb deletes an MDO variable:

ASSIGN [ OPT=VALUE ] MDO=variable-name

The following example produces the same result as the preceding example:

ASSIGN OPT=VALUE MDO=&abc.

Deleting an MDO variable using these verbs does not assign a null value to the
variable stem—it deletes the variable name and the names of all variables that begin
with the stem name from memory.

The following NCL procedure uses the ASSIGN verb to create an MDO variable,
&ABC., that is interpreted by the map $MSG:

zex1012n: PROCEDURE
   /* Creates an MDO variable using ASSIGN */
   SAY Creating an MDO variable with map $MSG
   ASSIGN MDO=&abc. OPT=VALUE MAP=$MSG
   SAY Creating fields in the MDO variable
   &abc.text                   = "A test message"
   &abc.msgattr.display.color  = RED
   &abc.msgattr.display.hlite  = NONE
   &abc.msgattr.display.intens = NORMAL
   SAY Displaying fields in the MDO variable
   SAY "Text is "&abc.text
   SAY "Color is "&abc.msgattr.display.color
   SAY "Hlite is "&abc.msgattr.display.hlite
   SAY "Intens is "&abc.msgattr.display.intens
   SAY Deleting an MDO variable
   DROP VARS=&abc.
   SAY &abc.
   SAY Attempting to display MDO variable
   SAY "Text is "&abc.text
   SAY "Color is "&abc.msgattr.display.color
   SAY "Hlite is "&abc.msgattr.display.hlite
   SAY "Intens is "&abc.msgattr.display.intens
END zex1012n

The NCL procedure creates a series of compound variables under the stem &ABC. and
displays their contents.  It then uses the DROP verb to delete &ABC..

The following screen shows the results of executing the procedure:

  (15:15) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1012N
   Creating an MDO variable with map $MSG
   Creating fields in the MDO variable
   Displaying fields in the MDO variable
   Text is A test message
   Color is RED
   Hlite is NONE
   Intens is NORMAL
   Deleting an MDO variable
   Attempting to display MDO variable

   Text is
   Color is
   Hlite is
   Intens is
   NNM1005 START ZEX1012N PROCESSING COMPLETE. NCLID 001560
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Assigning Source Data to an MDO Variable

You can assign source data to an MDO variable with the ASSIGN verb using the
following syntax:

ASSIGN [ OPT=VALUE ]
           MDO=variable-name [ MAP=map-name ]
           DATA=rest-of-statement

DATA=rest-of-statement

specifies data to assign to the MDO variable.

The following example assigns data to an MDO variable:

ASSIGN OPT=VALUE MDO=&abc. DATA=This is a statement

The following NCL procedure uses the ASSIGN verb to create an MDO variable,
&ABC., that is interpreted by the map $MSG:

zex1013n: PROCEDURE
   /* Assigns data to an MDO variable */
   ON ERROR DO
      SAY "An error has occurred"
      SAY "Line is "&SYS.ERROR.LINE
      SAY "Statement is "&SYS.ERROR.STMT_NAME
      SAY "Condition is "&SYS.ERROR.COND
      SAY "Assigning a new value to variable"
      IF &SYS.ERROR.STMT_NAME = "WRITE" THEN
         ASSIGN MDO=&abc. MAP=$MSG DATA=’00080001’x’WXYZ’
      RETRY
      END
   SAY Creating an MDO variable with map $MSG
   ASSIGN MDO=&abc. MAP=$MSG DATA=ABCD
   SAY "&ABC. is "&abc.
   WRITE DATA="&ABC.TEXT is "&abc.text
END zex1013n

The NCL procedure assigns data, ABCD, to the variable &ABC. and displays the
contents of the variable.  It then attempts to display the contents of the TEXT field,
using the WRITE verb.  However, Mapping Services raises an error because it cannot
interpret the values of &ABC.TEXT. according to the map definition for the TEXT
element.  The error handler then assigns data—’00080001’x’WXYZ’—to the variable.
(The characters 0008 represent a hexadecimal number with the length of the element
and the characters 0001 represent a hexadecimal number with the key value of the
element.)  The NCL procedure is then able to successfully display the contents of the
variable.

The following screen shows the results of executing the procedure:

  (07:30) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1013N
   Creating an MDO variable with map $MSG
   &ABC. is ABCD
   An error has occurred
   Line is 16
   Statement is WRITE
   Condition is VARIABLE_ERROR
   Assigning a new value to variable
   &ABC.TEXT is WXYZ
   NNM1005 START ZEX1013N PROCESSING COMPLETE. NCLID 001038
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Transferring Data From One MDO Variable to Another

You can transfer data from one MDO variable to another using the following syntax of
the ASSIGN verb:

ASSIGN [ OPT=VALUE ]
           MDO=variable-name [ MAP=map-name ]
       FROM
           MDO=variable-name

The following MSGPROC NCL procedure reads a message using the MSGREAD verb
into the MDO variable, &$MSG.:

zex1014n: PROCEDURE
   /* MSGPROC procedure that transfers data from */
   /* one MDO variable to another                */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         ASSIGN MDO=&abc. MAP=$MSG
         ASSIGN MDO=&abc. FROM MDO=&$MSG.
         SAY "&ABC.TEXT is "&abc.text
         SAY "&ABC.MSGATTR.DISPLAY.COLOR is "&abc.msgattr.display.color
         SAY "&ABC.MSGATTR.DISPLAY.HLITE is "&abc.msgattr.display.hlite
         SAY "&ABC.MSGATTR.DISPLAY.INTENS is "&abc.msgattr.display.intens
         SAY "&$MSG.MSGATTR.ALARM is "&$msg.msgattr.alarm
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1014n

The example uses the ASSIGN verb to create an MDO variable, &ABC., and to transfer
the data from &$MSG. to &ABC..

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:33) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1014N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   &ABC.TEXT is TEST An OPER NRD message
   &ABC.MSGATTR.DISPLAY.COLOR is WHITE
   &ABC.MSGATTR.DISPLAY.HLITE is NONE
   &ABC.MSGATTR.DISPLAY.INTENS is NORMAL
   &$MSG.MSGATTR.ALARM is 0
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001041
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Transferring Data From Ordinary Variables to an MDO Variable

You can transfer data from ordinary variables to an MDO variable using the following
syntax of the ASSIGN verb:

ASSIGN [ OPT=VALUE ]
           MDO=variable-name [ MAP=map-name ]
       FROM
         { ARGS [ RANGE=( start,end ) ] |
           VARS={ input-vars-list | variable* }
                [ RANGE=( start,end ) | GENERIC ] } ]

The following NCL procedure transfers data from the ordinary variables, &VAR1,
&VAR2, and &VAR3 to the MDO variables, &ABC1., &ABC2., &ABC3., and &ABC4.:

zex1015n: PROCEDURE
   /* Transfers data:  ordinary variables to MDO variables */
   &var1 = One  ; SAY "&VAR1 is "&var1
   &var2 = Two  ; SAY "&VAR2 is "&var2
   &var3 = Three; SAY "&VAR3 is "&var3
   ASSIGN MDO=&abc1. MAP=$MSG FROM VARS=&var*
   ASSIGN MDO=&abc2. MAP=$NCL FROM VARS=&var*
   ASSIGN MDO=&abc3. MAP=$SEC FROM VARS=&var*
   ASSIGN MDO=&abc4. MAP=$SPI FROM VARS=&var*
   SAY "&ABC1. ($MSG) is "&abc1.
   SAY "&ABC2. ($NCL) is "&abc2.
   SAY "&ABC3. ($SEC) is "&abc3.
   SAY "&ABC4. ($SPI) is "&abc4.
   SAY "$NCL is the exception:  display using map names"
   SAY "&ABC2.DATALIST.DATA{1} is "&abc2.datalist.data{1}
   SAY "&ABC2.DATALIST.DATA{2} is "&abc2.datalist.data{2}
   SAY "&ABC2.DATALIST.DATA{3} is "&abc2.datalist.data{3}
END zex1015n

The MDO variables use the maps $MSG, $NCL, $SEC, and $SPI, respectively, to
interpret the data.

The following screen shows the results of executing the procedure:

  (07:35) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1015N
   &VAR1 is One
   &VAR2 is Two
   &VAR3 is Three
   &ABC1. ($MSG) is OneTwoThree
   &ABC2. ($NCL) is  . . .  One .  Two .  Three
   &ABC3. ($SEC) is OneTwoThree
   &ABC4. ($SPI) is OneTwoThree
   $NCL is the exception:  display using map names
   &ABC2.DATALIST.DATA{1} is One
   &ABC2.DATALIST.DATA{2} is Two
   &ABC2.DATALIST.DATA{3} is Three
   NNM1005 START ZEX1015N PROCESSING COMPLETE. NCLID 001043
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The screen shows that data from the ordinary variables, &VAR1, &VAR2, and &VAR3,
is concatenated into a single string when you transfer it to the MDO variables &ABC1.,
&ABC3., and &ABC4..  It also shows that the map $NCL is the exception:  data
transferred to an MDO variable that uses the map $NCL is placed into separate fields,
not concatenated into a single string (this is the purpose of the map $NCL).

Transferring Data From an MDO Variable to Ordinary Variables

You can transfer data from an MDO variable to ordinary variables using the following
syntax of the ASSIGN verb:

ASSIGN [ OPT=VALUE ]
         { ARGS [ RANGE=( start,end ) ] |
           VARS={ output-vars-list | variable* }
                [ RANGE=( start,end ) |
                  BYNAME | GENERIC | MERGE | OVERLAY ] }
       FROM
           MDO=variable-name

The following NCL procedure transfers data from fields in the MDO variable &$MSG.
to ordinary variables:

zex1016n: PROCEDURE
   /* Transfers data:  MDO variables to ordinary variables */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         ASSIGN VARS=&color FROM MDO=&$MSG.MSGATTR.DISPLAY.COLOR
         ASSIGN VARS=&hlite FROM MDO=&$MSG.MSGATTR.DISPLAY.HLITE
         ASSIGN VARS=&intens FROM MDO=&$MSG.MSGATTR.DISPLAY.INTENS
         ASSIGN VARS=&alarm FROM MDO=&$MSG.MSGATTR.ALARM
         SAY "Color is "&color
         SAY "Hlite is "&hlite
         SAY "Intens is "&intens
         SAY "Alarm is "&alarm
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1016n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:37) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1016N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Color is WHITE
   Hlite is NONE
   Intens is NORMAL
   Alarm is 0
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001046
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Using the map $NCL allows you to assign data from ordinary variables to an MDO
variable and from an MDO variable to ordinary variables as required.  This allows you
to easily manipulate data to and from mapped user databases (UDBs), which store
ordinary variables in a form mapped by the map $NCL.  The following example
illustrates how to do so:

zex1017n: PROCEDURE
   /* Transfers data:  MDO variables to ordinary variables using $NCL */
      &var1 = One  ; SAY "&VAR1 is "&var1
      &var2 = Two  ; SAY "&VAR2 is "&var2
      &var3 = Three; SAY "&VAR3 is "&var3
      SAY "Transferring from ordinary variables to MDO variables"
      ASSIGN MDO=&abc2. MAP=$NCL FROM VARS=&var*
      SAY "&ABC2.DATALIST.DATA{1} is "&abc2.datalist.data{1}
      SAY "&ABC2.DATALIST.DATA{2} is "&abc2.datalist.data{2}
      SAY "&ABC2.DATALIST.DATA{3} is "&abc2.datalist.data{3}
      SAY "Transferring from an MDO variable to ordinary variables"
      ASSIGN VARS=&xyz* FROM MDO=&abc2.
      DO &i = 1 TO &SYS.VARCNT
         SAY "&XYZ"&i" is "&xyz&i
         END
END zex1017n

The following screen shows the results of executing the procedure:

  (07:42) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1017N
   &VAR1 is One
   &VAR2 is Two
   &VAR3 is Three
   Transferring from ordinary variables to MDO variables
   &ABC2.DATALIST.DATA{1} is One
   &ABC2.DATALIST.DATA{2} is Two
   &ABC2.DATALIST.DATA{3} is Three
   Transferring from an MDO variable to ordinary variables
   &XYZ1 is One
   &XYZ2 is Two
   &XYZ3 is Three
   NNM1005 START ZEX1017N PROCESSING COMPLETE. NCLID 001048
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The screen shows that data from the ordinary variables &VAR1, &VAR2, and &VAR3
is placed into separate fields in a mapped user database (UDB) record, not
concatenated into a single string.  It shows that you can read the data from a mapped
UDB record into ordinary variables.

Switching Maps

You can switch the map that is used to interpret the data in an MDO variable by using
the following syntax of the ASSIGN verb:

ASSIGN [ OPT=VALUE ] MDO=variable-name MAP=map-name

The following example directs Mapping Services to use the map $SEC (rather than the
usual map $MSG) to interpret data in the MDO variable &$MSG.:

ASSIGN OPT=VALUE MDO=&$MSG. MAP=$SEC

Note Switching maps is not recommended since it may produce unpredictable results.

Format 2 of the ASSIGN
Verb

Format 2 of the ASSIGN verb permits many values to follow the OPT keyword.  This
syntax is used exclusively with MDO variables and maps.

The following table summarizes the meaning of the MAPNAME, MAPELEMENTS,
MAPFIELDS, MDOELEMENTS, MDOFIELDS, and MDOCOUNTS values, which are
used in Format 2 of the ASSIGN verb:

Value Description

MAPNAME Obtains the name of the map used to interpret an MDO variable.

MAPELEMENTS Obtains the names of the elements defined in a map at the specified level.

MAPFIELDS Obtains the names of the groups or fields defined in a map at the specified
level.

MDOELEMENTS Obtains the names of the elements present in an MDO variable at the specified
level.

MDOFIELDS Obtains the names of the groups or fields present in an MDO variable at the
specified level.

MDOCOUNTS Obtains the number of occurrences of the element, group, or field present in an
MDO variable at the specified level.

Obtaining the Map Name for an MDO Variable

You can obtain the name of the map used to interpret an MDO variable using the
following syntax of the ASSIGN verb:

ASSIGN OPT=MAPNAME
           { ARGS [ RANGE=( start,end ) ] |
             VARS={ output-vars-list | variable* }
                  [ RANGE=( start,end ) ] }
       FROM
             MDO=variable-name

The following MSGPROC NCL procedure reads a message and uses the ASSIGN verb
to obtain the name of the map used to interpret the MDO variable into which the
message is placed:

zex1018n: PROCEDURE
   /* MSGPROC procedure to trap a message and display */
   /* its map name                                    */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         ASSIGN OPT=MAPNAME VARS=&mapname FROM MDO=&$MSG.
         SAY "Map name is "&mapname
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1018n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:48) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1018N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Map name is $MSG
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001052
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Obtaining Element Names in a Map

You can obtain the names of the elements defined in a map at the specified level using
the following syntax of the ASSIGN verb:

ASSIGN OPT=MAPELEMENTS
           { ARGS [ RANGE=( start,end ) ] |
             VARS={ output-vars-list | variable* }
                  [ RANGE=( start,end ) ] }
       FROM
             MDO=variable-name

The following MSGPROC NCL procedure reads a message and uses the ASSIGN verb
to obtain element names:

zex1019n: PROCEDURE
   /* MSGPROC procedure to trap a message and display */
   /* element names in a map                          */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         ASSIGN OPT=MAPELEMENTS VARS=&mapelements*,
            FROM MDO=&$MSG.
         SAY "Elements in map $MSG are"
         DO &i = 1 TO &SYS.VARCNT
            SAY &mapelements&i
            END
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1019n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:49) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1019N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Elements in map $MSG are
   CNM
   DOMID
   EVENT
   MAPNAME
   MSGATTR
   PPOCNTL
   PREFIX
   RESOURCE
   SOURCE
   SPECIAL
   SPI
   TEXT
   USERMDO
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001055

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Obtaining Group and Field Names in a Map

You can obtain the names of the groups or fields defined in a map at the specified level
using the following syntax of the ASSIGN verb:

ASSIGN OPT=MAPFIELDS
           { ARGS [ RANGE=( start,end ) ] |
             VARS={ output-vars-list | variable* }
                  [ RANGE=( start,end ) ] }
       FROM
             MDO=variable-name

If the MDO operand specifies an element that has no defined groups or fields, then no
data is returned.  If the MDO operand specifies a group name, only those fields
defined within that group are returned.  If the MDO operand specifies a group that
contains no further fields, or a single field name, then no data is returned.

The following MSGPROC NCL procedure reads a message and uses the ASSIGN verb
to obtain field names in a map:

zex1020n: PROCEDURE
   /* MSGPROC procedure to trap a message and display */
   /* field names in a map                            */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         ASSIGN OPT=MAPFIELDS VARS=&mapfields*,
            FROM MDO=&$MSG.MSGATTR.DISPLAY
         SAY "Fields in map $MSG.MSGATTR.DISPLAY are"
         DO &i = 1 TO &SYS.VARCNT
            SAY &mapfields&i
            END
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1020n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:51) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1020N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Fields in map $MSG.MSGATTR.DISPLAY are
   BKGCOLOR
   BKGCOLOR#
   COLOR
   COLOR#
   HLITE
   HLITE#
   INTENS
   INTENS#
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001058
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Obtaining Element Names in an MDO Variable

You can obtain the names of the elements present in an MDO variable at the specified
level using the following syntax of the ASSIGN verb:

ASSIGN OPT=MDOELEMENTS
           { ARGS [ RANGE=( start,end ) ] |
             VARS={ output-vars-list | variable* }
                  [ RANGE=( start,end ) ] }
       FROM
             MDO=variable-name

Only the names of those elements present within the specified MDO variable are
returned.

The following MSGPROC NCL procedure reads a message and uses the ASSIGN verb
to obtain the names of the elements that are present:

zex1021n: PROCEDURE
   /* MSGPROC procedure to trap a message and display */
   /* elements names in an MDO variable               */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         ASSIGN OPT=MDOELEMENTS VARS=&mdoelements*,
            FROM MDO=&$MSG.
         SAY "Elements under &$MSG. are"
         DO &i = 1 TO &SYS.VARCNT
            SAY &mdoelements&i
            END
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1021n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:53) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1021N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Elements under &$MSG. are
   TEXT
   MSGATTR
   SOURCE
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001061
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Obtaining Field Names in an MDO Variable

You can obtain the names of the fields present in an MDO variable at the specified
level using the following syntax of the ASSIGN verb:

ASSIGN OPT=MDOFIELDS
           { ARGS [ RANGE=( start,end ) ] |
             VARS={ output-vars-list | variable* }
                  [ RANGE=( start,end ) ] }
       FROM
             MDO=variable-name

Only the names of those fields present within the specified MDO variable are
returned.

The following MSGPROC NCL procedure reads a message and uses the ASSIGN verb
to obtain the names of the fields that are present:

zex1022n: PROCEDURE
   /* MSGPROC procedure to trap a message and display */
   /* field names in an MDO variable                  */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         ASSIGN OPT=MDOFIELDS VARS=&mdofields*,
            FROM MDO=&$MSG.MSGATTR.FLAGS
         SAY "Fields under &$MSG.MSGATTR.FLAGS are"
         DO &i = 1 TO &SYS.VARCNT
            SAY &mdofields&i
            END
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1022n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:55) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1022N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Fields under &$MSG.MSGATTR.FLAGS are
   CLEAR
   CMDECHO
   CMDLINE
   CONVPFK
   EXPRESS
   FIRST
   INTQ
   LAST
   LOG
   LOGP
   MSG
   MSGP
   NRD
   NRDRET
   SCAN
 MSG QUEUED ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Obtaining Information About Repeated Elements in an MDO Variable

You can obtain information about repeated elements in an MDO variable at the
specified level using the following syntax of the ASSIGN verb:

ASSIGN OPT=MDOCOUNTS
           { ARGS [ RANGE=( start,end ) ] |
             VARS={ output-vars-list | variable* }
                  [ RANGE=( start,end ) ] }
       FROM
             MDO=variable-name

The following MSGPROC NCL procedure reads a message and uses the ASSIGN verb
to obtain information about variable components:

zex1023n: PROCEDURE
   /* MSGPROC procedure to display information about */
   /* variable components in an MDO variable         */
   DO FOREVER
      MSGREAD VARS=&word*
      IF &word1 = "TEST" THEN DO
         &$msg.msgattr.fld{1}.offset = 8
         &$msg.msgattr.fld{1}.length = 8
         &$msg.msgattr.fld{1}.color = "TURQUOISE"
         &$msg.msgattr.fld{1}.hlite = "BLINK"
         &$msg.msgattr.fld{1}.intens = "HIGH"
         &$msg.msgattr.fld{2}.offset = 17
         &$msg.msgattr.fld{2}.length = 7
         &$msg.msgattr.fld{2}.color = "BLUE"
         &$msg.msgattr.fld{2}.hlite = "NONE"
         &$msg.msgattr.fld{2}.intens = "NORMAL"
         ASSIGN OPT=MDOCOUNTS VARS=&mdocounts,
            FROM MDO=&$MSG.MSGATTR.FLD
         SAY "Occurrences of &$MSG.MSGATTR.FLD is "&mdocounts
         MSGCONT
         END
      ELSE
         MSGCONT
   END
END zex1023n

After following the steps outlined earlier in this section to type this NCL procedure
and install it as your active MSGPROC procedure, execute the ZEX1002N NCL
procedure to send a message to your OCS window.  The following screen shows the
results of executing ZEX1002N with MSGPROC active:

  (07:57) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1023N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   START ZEX1002N
   TEST An OPER NRD message
   Occurrences of &$MSG.MSGATTR.FLD is 2
   NNM1005 START ZEX1002N PROCESSING COMPLETE. NCLID 001067
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

